Asset Pricing

1. Version vom 6. Juni 2005

Einleitung

Das Ziel dieses Dokuments ist es, in kurzer und prägnanter Form eine Anwendung von *Mathematica* im Finanzbereich zu zeigen. Als Beispiel wird die Berechnung der Wertenwicklung verschiedener Wertschrifttten (Bargeld, Aktie, Option auf eine Aktie) behandelt. Es ist klar, dass zu jedem Thema noch vieles gesagt werden könnte; insbesondere könnten die Mathematik präziser und rigoroser formuliert und die Voraussetzungen für die aufgeführten Behauptungen und Sätze genauer spezifiziert werden. Aus Platzgründen wird jedoch darauf verzichtet, wie auch auf die Erklärung vieler im Finanzbereich bestens bekannter Ausdrücke wie Aktie, Put, Call etc.

Das vorliegende PDF Dokument wurde mit *Mathematica* (als NB Datei) geschrieben und als PDF Datei exportiert. Es wird ausgiebig Gebraucht gemacht von den Möglichkeiten von *Mathematica* als Dokumentationssystem mit Style Sheets, als Werkzeug zur Lösung von Differenzialgleichungen, als Problemlöser im Bereich der Statistik, als Graphik Engine sowie als Programmiersprache. Es ist augenfällig, wie - auf Grund der Vielzahl von eingebauten Funktionen - oft mit wenig Code (es werden nur die Standard *Mathematica* Funktionen und Packages verwendet) ein grosses Problem gelöst und die Lösung anschaulich dargestellt werden kann. Die für den Neuling etwas gewöhnungsbedürftige *Mathematica* Syntax (insbesondere die funktionale Programmierung) wird aus Platzgründen nicht erklärt. Der Interessierte kann sich unter www.wolfram.com ausgiebig informieren. Zur besseren Übersicht wird in diesem Dokument *Mathematica* Input (Formeln und Befehle) in blau und der von *Mathematica* retournierte Ausdruck in grün dargestellt.

Der Inhalt dieses Dokuments enthält die folgenden Kapitel:

Im ersten Kapitel wird als Einstieg ein sehr einfacher **Cash Account** behandelt, dessen Wert durch Zins und Zinseszins zunimmt und dessen Preisentwicklung durch eine (einfache) Gewöhnliche Differenzialgleichung (ODE) beschrieben werden kann.

Im zweiten Kapitel wird die Preisentwicklung einer **Aktie** diskutiert. Diese Preisentwicklung ist zusätzlich zu deterministischen Termen auf Grund von vielen zufälligen Einflüssen auch durch nicht-deterministische Terme, die durch Brown'sche Bewegungen simuliert werden, bestimmt. Dies führt auf eine stochastische Differenzialgleichung (SDE), die mittels Monte-Carlo Verfahren simuliert oder mittels Ito Calculus symbolisch (für eine gegebenen Trajektorie der Brown'schen Bewegung) berechnet werden kann.

Im dritten Kapitel wird auf die noch um eine Stufe schwierigere Berechnung einer **Europäischen Option (Put oder Call)** mit einer Aktie als underlying eingegangen. Bei der Analyse der Preisentwicklung wird die Black-Scholes Partielle Differenzialgleichung (PDE) hergeleitet und durch eine Variablen Transformation in eine (in der Physik bestens bekannte) Wärmeleitungsgleichung umgewandelt und gelöst. Lösungen der Black-Scholes PDE werden diskutiert.

Diese Kapitel geben einen ersten Eindruck über die Fähigkeiten von *Mathematica*. Mit Kapiteln über **American Style Stock Options**, **Optimal Portfolio Rules** sowie **Advanced Trading Strategies** könnte dieses Dokument noch ausgebaut werden.

Als Quelle für diese kurze Zusammenstellung diente u.a. das Buch "Computational Financial Mathematics using *Mathematica*® –Optimal Trading in Stocks and Options" (2002), von Srdjan Stojanovic, University of Cincinnati.

Cash - Bargeld

Der Wert eines Cash Account nimmt exponentiell mit der Zeit und dem (konstanten) Zins r zu und wird durch folgende Differenzialgleichung beschrieben:

$y'(t)=r\,y(t)$

$y(0)=\mathrm{y0}$

Die Lösung lautet:

$e^{r\,t}\,\mathrm{y0}$

Gewöhnliche Differenzialgleichung

Cash Accounts, die (praktisch) risikofrei sind, lassen sich mathematisch sehr einfach beschreiben:

Angenommen ein Investor hat y0 Euros in einem Cash Account, der r 100% Zins pro Jahr (stetige Verzinsung) einbringt. Dann entwickelt sich das Guthaben gemäss der (sehr einfachen) Gewöhnlichen Differenzialgleichung (ODE: Ordinary Differential Equation)

$$y'(t) = r(t)\,y(t) \tag{1}$$

mit der Anfangsbedingung

$$y(0) = \mathrm{y0} \tag{2}$$

Die Lösung der Differenzialgleichung kann am einfachsten mit der *Mathematica* Funktion DSolve gelöst werden:

```
sol = DSolve[{y'(t) == r y(t), y(0) == y0}, y(t), t][[1, 1, 2]]
```

$e^{r\,t}\,\mathrm{y0}$

Das Guthaben nimmt also exponentiell zu. Dieses Verhalten kann auch anschaulich mit einem Plot dargestellt werden. Im folgenden Beispiel wird dargestellt, wie das Kapital von 100 Euro in 10 Jahren mit verschiedenen konstanten Zinssätzen r zunimmt.

```
Plot[Evaluate[(sol /. {r → #1, y0 → 100} &) /@ {0.03, 0.05, 0.07}],
  {t, 0, 10}, PlotRange → {80, 200}, AxesLabel → {"Zeit", "Guthaben"}, PlotStyle → {Red, Green, Blue}];
```



Im vorigen Beispiel galt ein gleichbleibender Zins. Zur Lösung dieser Aufgabe wäre die Verwendung von *Mathematica* sicher nicht notwendig. In schwierigeren Fällen kann *Mathematica* jedoch einige Routinearbeiten abnehmen. Dies soll an Hand eines zeitabhängigen Zinses gezeigt werden.

Im ersten Schritt wird der Zins definiert (und zur Veranschaulichung geplottet):

$$r(t\_) := \begin{cases} 0.06 & t < 3 \\ 0.02 & 3 \le t < 7 \\ 0.06 & 7 \le t \end{cases};$$

Plot[$r(t)$, {t, 0, 10}, PlotRange → {0, 0.10}, AxesLabel → {"Zeit", "Zins"}, PlotStyle → {Red}];



Im zweiten Schritt wird die Differenzialgleichung gelöst

sol = DSolve[{$y'(t) = r(t)\, y(t)$, $y(0) =$ y0}, $y(t)$, t]

$$\left\{\left\{y[t] \to \begin{cases} e^{0.06\grave{}\, t}\, y0 & -\infty < t \le 3 \\ e^{0.02\grave{}\, t}\, (0.\grave{} + 1.1274968515793757\grave{}\, y0) & 3 < t \le 7 \\ e^{0.06\grave{}\, t}\, (0.\grave{} + 0.8521437889662115\grave{}\, y0) & \text{True} \end{cases}\right\}\right\}$$

und geplottet

p1 = Plot[$y(t)$ /. sol /. y0 → 100, {t, 0, 10}, PlotRange → {80, 160}, AxesLabel → {"Zeit", "Guthaben"},
PlotStyle → {Red}];



Stock - Aktie

Die Preisentwicklung einer Aktie wird durch eine Stochastische Differenzialgleichung beschrieben

$$d y(t) = a\ y(t)\ dt + \sigma\ y(t)\ dB(t)$$

$$y(t_0) = y_0$$

Über die effektive Preisentwicklung können nur Wahrscheinlichkeitsaussagen gemacht werden.

Im Vergleich zu (praktisch) risikofreien Instrumenten wie Cash Accounts sind Investitionen in risikoreiche Wertschriften wie z.B. Aktien mit grösseren Unsicherheiten behaftet.

Aktien haben einen Preis auf dem offenen Markt, der sich praktisch kontinuierlich ändert. Diese Fluktuationen des Aktienpreises stellen die konstante Suche nach einem fairen Preis dar. Zusätzlich zu diesen (zufälligen) Fluktuationen gibt es eine mehr oder weniger kontinuierliche (in der Zeit) Zunahme oder Abnahme des Werts, die auf das wirtschaftliche Umfeld oder firmenspezifische Faktoren zurückzuführen sind.

Stochastiche Differenzialgleichung

Die Berechnung des Werts einer Aktie ist aus den angeführten Gründen nicht so einfach wie die Berechnung des Werts von Cash. Die Änderung des Werts ist nicht nur von der Zeit und dem momentanten Wert der Aktie abhängig. In komplizierter Art und Weise hängt die Änderung ausserdem von vielen weiteren Dingen ab (Inflationsrate, Zins, Arbeitslosigkeit, Währungskurse, etc.), die nicht mit genügender Genauigkeit modelliert werden können.

Aus diesem Grund wird der Differenzialgleichung ein Zufallselement hinzugefügt, das diese nicht deterministischen Terme enthalten soll. Dies führt auf den folgenden Ansatz für die zeitliche Änderung des Werts einer Aktie:

$$\begin{aligned} d\,y(t) &= a(t,\ y(t))\,d\,t + \sigma(t,\ y(t))\,d\,B(t) \\ y(t_0) &= y_0 \end{aligned} \tag{3}$$

wo a(t,y(t)) dt den deterministischen Teil und σ(t,y(t)) dB(t) den zufälligen Teil beschreibt. $dB(t)$ ist dabei das "Differential" der Brownschen Bewegung $B(t)$, und $\sigma(t,\ y)$ eine gegebene Funktion ($\frac{\sigma(t,y(t))}{y(t)}$ wird Volatilität genannt). Mit Brown'scher Bewegung ist gemeint, dass die $d\,B(t)$'s (unabhängige) normal verteilte Zufallsvariablen sind, mit Mittelwert 0 und Standard Abweichung $\sqrt{d\,t}$ (Varianz $d\,t$), i.e.,

$$d\,B(t) \sim N\left(0,\ \sqrt{d\,t}\right).$$

Die obige Gleichung wird Stochastische Differenzialgleichung (SDE: stochastic differential equation) oder präziser Stochastische Gewöhnliche Differenzialgleichung genannt.

Zusätzlich wird im obigen Gleichungssystem auch noch die Randbedingung - d.h. der anfängliche Wert der Aktie $y(t_0)$ - festgelegt.

Brown'sche Bewegung

Zuerst soll die Brown'sche Bewegung etwas Genauer untersucht werden. Dazu wird im Folgenden die Funktion "BrownianMotion" definiert und verwendet:

Als Input verlangt sie die Startzeit (t0), die Endzeit (t1), den Anfangswert (y0) sowie die Anzahl der Schritte (K) der Brown'schen Bewegung.

Bei der Berechnung, die auch auf das Statistik Paket zurückgreift, werden zuerst die Schrittweite (dt), dann die Liste der einzelnen Schritte (dB, wobei Schritte aus einer Normalverteilung mit Varianz dt stammen) und schlussendlich mit FoldList auch noch die Listen mit den Zeitpunkten und den aufsummierten Schritten (d.h. die Trajektorie) zu diesen Zeitpunkten berechnet.

Diese drei Listen werden von der Funktion als Output retourniert.

```
Needs["Statistics`NormalDistribution`"];
BrownianMotion[t0_, t1_, y0_, K_] := Module[{dt, dB},
  dt = N[(t1 - t0)/K];
  {dB = Table[Random[NormalDistribution[0, √dt]], {K}],
   FoldList[{dt + #1⟦1⟧, #1⟦2⟧ + #2} &, {0, 0}, dB]}
 ];
```

Im Folgenden werden vier solcher Trajektorien berechnet. Jeder Aufruf dieser Funktion gibt auf Grund des Aufrufs von "Random" in der Funktion BrownianMotion eine andere Trajektorie. Mit "Interpolation" wird zum Plotten zwischen den Punkten der Trajektorie linear interpoliert.

```
Plot[Evaluate[Table[Interpolation[BrownianMotion[0, 1, 0, 100][[2]],
     InterpolationOrder → 1][s], {4}]], {s, 0, 1}, PlotStyle → {Red, Green,
   Blue, Black}, AxesLabel → {"Zeit", ""}];
```



Monte-Carlo Lösung der SDE

Nachdem wir gesehen haben, wie die Brown'sche Bewegung programmiert werden kann, soll nun die SDE gelöst werden. Lösen heisst hier, ein Verfahren zu finden, um den Verlauf des Aktienpreises (je nach Verlauf der Brown'schen Bewegung) zu berechnen. Die SDE lautet in der diskretisierten Darstellung

$$y_{i+1} = y_i + a(t_i, y_i)\, dt + \sigma(t_i, y_i)\, dB_i \tag{4}$$

Zur Lösung der SDE wird die Funktion "SDESolver" verwendet. Im Vergleich zu "BrownianMotion" wird hier nicht nur die Schrittlänge aufsummiert, sondern alle in der obigen Gleichung gegebenen Terme. Beim Aufruf der Funktion muss auch die Drift $a(t, y)$ und die nicht-deterministische Funktion $\sigma(t, y)$ eingegeben werden. Die Funktion "SDESolver" wird folgendermassen definiert:

```
Needs["Statistics`NormalDistribution`"];
SDESolver[aFunc_, σFunc_, t0_, t1_, y0_, K_] := Module[{dt, G},
  dt = N[(t1 - t0)/K];
  G[{t_, y_}, db_] := {t + dt, y + aFunc[t, y] dt + σFunc[t, y] db};
  FoldList[G, {t0, y0}, Table[Random[NormalDistribution[0, √dt]], {K}]]
 ];
```

Die einfachste SDE für eine Aktienpreis Entwicklung stellt die spezielle Wahl der Funktionen $a(t_i, y_i) = a\, y_i$ und $\sigma(t_i, y_i) = \sigma\, y_i$ dar:

$$\begin{aligned} dy(t) &= a\, y(t)\, dt + \sigma\, y(t)\, dB(t) \\ y(t_0) &= y_0 \end{aligned} \tag{5}$$

Experimentell (Monte-Carlo) kann diese Gleichung - mit Hilfe der oben definierten Funktion "SDESolver" - gelöst werden. Im Folgenden werden a und σ gesetzt sowie 10 mögliche Aktienpreisverläufe berechnet und geplottet.

```
aFunc[t_, y_] := .3 y;
σFunc[t_, y_] := .1 y;
t0 = 0; t1 = 2; y0 = 100; K = 1000;
Plot[
 Evaluate[Table[Interpolation[SDESolver[aFunc , σFunc , t0, t1, y0, K],
     InterpolationOrder → 1][t], {10}]],
 {t, t0, t1}, PlotRange → {0, Automatic}, PlotStyle → {Red, Green, Blue,
   Magenta, Black, Cyan}, AxesLabel → {"Zeit", "Aktienpreis"}];
```



Symbolische Lösung der SDE

Nach der numerischen Monte-Carlo Lösung soll noch auf die symbolische Lösung der SDE eingegangen werden.

$$\begin{aligned} d\,y(t) &= a\,y(t)\,d\,t + \sigma\,y(t)\,d\,B(t) \\ y(t_0) &= y_0 \end{aligned} \tag{6}$$

Bei Stochastischen Differenzialgleichungen ist zu berücksichtigen, dass nicht die aus der Analysis gewohnten Regeln (z.B. Kettenregel, Produktregel, Integration) zu verwenden sind, sondern die für Stochastische Gleichungen angepassten (z.B. Ito Kettenregel, Ito Integration). Auf diese Details wird hier jedoch nicht näher eingegangen.

Mit dem Ansatz für den zeitabhängigen Preis y(t)

$$z = \log(y) \tag{7}$$

und Ausnutzung der Ito Kettenregel

$$(d\,y)^2 = \sigma^2\,y^2\,d\,t \tag{8}$$

kann man ableiten, dass

$$d\,z = d\log(y) = \frac{1}{y}\,d\,y + \frac{1}{2}\,\frac{-1}{y^2}\,(d\,y)^2 = a\,d\,t + \sigma\,d\,B - \frac{1}{2\,y^2}\,\sigma^2\,y^2\,d\,t = \left(a - \frac{1}{2}\,\sigma^2\right)\,d\,t + \sigma\,d\,B \tag{9}$$

Es fällt auf, dass die Drift von z nicht gleich der Drift von y ist. Stochastische Integration liefert dann

$$z(t) = t\left(a - \frac{\sigma^2}{2}\right) + \sigma\,(B(t) - B(0)) + z(0) = t\left(a - \frac{\sigma^2}{2}\right) + \sigma\,B(t) + \log(\mathrm{y0}). \tag{10}$$

Exponenzieren liefert schliesslich

$$y(t) = e^{z(t)} = e^{\left(a - \frac{\sigma^2}{2}\right)t + \sigma\,B(t) + \log(\mathrm{y0})} = \mathrm{y0}\,e^{\left(a - \frac{\sigma^2}{2}\right)t + \sigma\,B(t)}. \tag{11}$$

Diese Gleichung ist nicht besonders nützlich für die Berechnung von y(t), da die Brown'sche Bewegung B(t) nicht gemessen werden kann und y(t) ja sowieso vom Markt geliefert wird.

Die Gleichung liefert jedoch Grenzen für den Verlauf des Preises. Man sieht auch schnell, dass der Median den Verlauf $\mathrm{y0}\,e^{\left(a - \frac{\sigma^2}{2}\right)t}$ hat und etwas tiefer als der Durchschnittspreis ($\mathrm{y0}\,e^{a\,t}$) liegt.

Im untenstehenden Plot werden einige (mögliche) Preisentwicklungen (grün), die gegebenen Grenzen (±σ innerhalb dessen 68.3% der Werte liegen sollten, ±2 σ mit 95.5% und ±3 σ mit 99.7%) in schwarz, der Median (blau) sowie der Durchschnitt des Preises (rot) eingezeichnet.

```
y0 = 70; a = .5; σ = .6; t0 = 0; t1 = 1; T = t1 - t0; K = 200;
aFunc[t_, y_] := a y;
σFunc[t_, y_] := σ y;
Z[a_, σ_, b_, t_] = y0 ⅇ^((a - σ^2/2) t + b σ Sqrt[t]);
Show[
 Plot[Evaluate[Table[Interpolation[SDESolver[aFunc , σFunc , t0, t1, y0,
       K], InterpolationOrder → 1][t], {50}]], {t, t0, t1},
  PlotRange → {0, 400}, PlotStyle → Green, DisplayFunction → Identity],
 Plot[Evaluate[Table[Z[a, σ, b, t], {b, -3, 3}]], {t, 0, T},
  PlotStyle → ReplacePart[Table[RGBColor[0, 0, 0], {6}], RGBColor[0, 0, 1],
    4], DisplayFunction → Identity],
 Plot[y0 ⅇ^(a t), {t, 0, T},
  PlotStyle → RGBColor[1, 0, 0], DisplayFunction → Identity],
 DisplayFunction → $DisplayFunction, AxesLabel → {"Zeit", "Aktienpreis"}];
```



Mehrere Aktien

Die bisherige Untersuchung ging von einer einzelnen Aktie aus. In ähnlicher (jedoch etwas komplizierterer) Weise können auch mehrere Aktien behandelt werden. Bei mehreren Aktien muss die Multinormal Verteilung verwendet sowie die Kovarianz zwischen den Preisen berücksichtigt werden.

European Style Stock Options

Der Wert einer Stock Option (V) am Verfallstag (T, expiration date) hängt sowohl vom (fixen) Ausübungspreis (k, exercise price) der Option und vom (erst am Verfallstag bekannten) Wert (S) der Aktie zu jenem Zeitpunkt ab. Mathematisch

$$\begin{aligned}&\text{Call}: \; V(T,\,S) = \text{Max}(S(T) - k,\,0)\\&\text{Put}: \; V(T,\,S) = \text{Max}(k - S(T),\,0)\end{aligned} \tag{12}$$

Die interessante Frage ist nun, wie sich der Wert der Option vor dieser Zeit entwickelt, unter Berücksichtigung der Tatsache, dass der Wert der Option vom Aktienpreis abhängt und die Aktienpreisentwicklung nicht-deterministisch (siehe σ) ist. Die Antwort lautet:

Die Preisentwicklung $V(t)$ einer European Style Stock Option wird durch die Black–Scholes Partielle Differenzialgleichung beschrieben mit der dazugehörigen Randbedingung (hier nur für Call, Bedeutung der Symoble $s.u.$):

$$\frac{1}{2}\,S^2\,\frac{\partial^2 V(t,S)}{\partial S^2}\,\sigma(t,S)^2+(r(t)-D_0(t,S))\,S\,\frac{\partial V(t,S)}{\partial S}+\frac{\partial V(t,S)}{\partial t}-r(t)\,V(t,S)=0;$$

$$V(T,S)=\text{Max}(S-k,0)$$

Über die genauen Voraussetzungen für diese PDE kann man sich in der Literatur informieren.

Die Lösung dieser Gleichung ist recht komplex und lautet für den (einfachen) Fall einer Aktie ohne Dividende und für konstanten Zins r:

$$\frac{1}{2}\left(S+S\,\text{Erf}\left[\frac{-(t-T)\,(2\,r+\sigma^2)+2\,\text{Log}[\frac{S}{k}]}{2\,\sqrt{2}\,\sqrt{(-t+T)\,\sigma^2}}\right]+e^{r\,(t-T)}\,k\left(-2+\text{Erfc}\left[\frac{\sqrt{(-t+T)\,\sigma^2}\,\left(-\frac{1}{2}+\frac{r}{\sigma^2}+\frac{\text{Log}[\frac{S}{k}]}{(-t+T)\,\sigma^2}\right)}{\sqrt{2}}\right]\right)\right)$$

Wie man unschwer sieht, ist die Preisentwicklung einer European Style Stock Option um einiges komplizierter als die Preisentwicklung eines Cash Accounts oder einer Aktie.

Call Payoff

Um einen ersten Eindruck zu bekommen soll der mögliche Payoff einer Call Option betrachtet werden. Die Preisentwicklung der dazugehörigen Aktie soll wie im letzten Abschnitt durch die SDE

$$\begin{aligned}&d\,y(t) = a(t,\;y(t))\,d\,t + \sigma(t,\;y(t))\,d\,B(t)\\&y(t_0) = y_0\end{aligned} \tag{13}$$

beschrieben und mit der dortigen Funktion "SDESolver" berechnet werden.

Im Folgenden werden 100 Versuche durchgeführt, wobei zuerst alle Input Parameter festgelegt und dann SDESolver aufgerufen und mit Max die Auszahlungen (Stock Preis minus Ausübungspreis, oder 0) berechnet werden. Mit Sort wird die Liste sortiert und die Auszahlungen werden in aufsteigender Form geplottet.

```
k = 80;
a = .5; σ = .5; y0 = 75; t0 = 0; t1 = 3/12; T = t1 - t0; K = 200;

aFunc[t_, y_] := a y; σFunc[t_, y_] := σ y;
(Max[#1 - k, 0] &) /@ (tabl = Table[SDESolver[aFunc, σFunc, t0, t1, y0, K][[
      -1, 2]], {500}]);
ListPlot[Sort[%], PlotStyle → {PointSize[0.017], Red}, FrameLabel →
   {"Versuch", "Payoff"}, Frame → True, GridLines → Automatic];
```



Man sieht, dass in vielen Fällen die Call Option verfällt, aber auch dass in einigen Fällen ein grosser Gewinn resultiert, da (wie die späteren Berechnungen zum Marktpreis der Option zeigen werden) der Kaufpreis viel kleiner ist als die grösstmöglichen Auszahlungen.

Was ist jedoch der faire Preis für eine solche Option? Ein möglicher Ansatz bestünde darin, für sehr viele mögliche Aktienpreisverläufe den Payoff zu berechnen und den Wert auf den Barwert abzudiskontieren. Dies würde für die obigen Payoffs (tabl) folgenden Wert liefern, wobei die Funktion "Max" (wie oben) den Payoff des Calls und die Funktion "Mean" den Mittelwert berechnet, und mit dem Zins r abdiskontiert wird.

```
r = 0.05;
Mean[Max[# - k, 0] & /@tabl] Exp[-T r]

10.360456645938209`
```

Die richtige Antwort ist jedoch etwas komplizierter und wurde von Black und Scholes Anfang der 1970er Jahre erarbeitet. Auf diese Gleichung wird in den nächsten Abschnitten eingegangen.

Herleitung der Black-Scholes PDE

Bei der Herleitung wird auch die Möglichkeit berücksichtigt, dass Dividenden (D) für die Aktien ausbezahlt werden. Die Aktien Preisentwicklung S(t) wird dann durch folgende Differenzialgleichung (siehe auch weiter oben) beschrieben:

$$\begin{aligned} dS(t) &= S(t)\,((a(t, S(t)) - D_0(t, S(t)))\,dt + \sigma(t, S(t))\,dB(t)) \\ S(0) &= S_0 \end{aligned} \tag{14}$$

Es wird angenommen, dass der Investor eine (Put) Option mit dem (zu bestimmenden) Wert V, eine auf Grund des Hedging zu bestimmende Anzahl von Aktien Δ sowie einen Cash Account C besitzt, also

$$X(t) = V(t, S(t)) - \Delta(t, S(t))\,S(t) + C(t) \tag{15}$$

Unter Verwendung des Ito Calculus kann berechnet werden, dass sich die Position X(t) folgendermassen ändert:

$$\begin{aligned} dX(t) = {} & V_t(t, S(t))\,dt + V_S(t, S(t))\,dS(t) + \frac{1}{2}\,V_{S,S}(t, S(t))\,(dS(t))^2 - \Delta(t, S(t))\,dS(t) + \\ & r(t)\,C(t)\,dt - \Delta(t, S(t))\,S(t)\,D_0(t, S(t))\,dt \end{aligned} \tag{16}$$

Das Ziel der Hedging Strategie ist nun, soviele Aktien (Δ) zu halten, dass das Risiko aus dieser Gleichung eliminiert wird. Der erste Ansatz lautet:

$$\Delta(t, S) = V_S(t, S) = \frac{\partial V(S(t), t)}{\partial S} \tag{17}$$

woraus folgt, dass:

$$d X(t) = V_t(t, S(t))\, d t + \frac{1}{2} V_{S,S}(t, S(t))\, S(t)^2\, \sigma(t, S(t))^2\, d t + r(t)\, C(t)\, d t - V_S\, (t, S(t))$$
$$S(t)\, D_0(t, S(t))\, d t \tag{18}$$

Man sieht, dass die Randomness, die im Aktienpreis S steckt, noch nicht vollständig eliminiert wurde. Deshalb wird noch zusätzlich verlangt, dass:

$$d X(t) = r(t)\, X(t)\, d t \tag{19}$$

Dies bedeutet, dass der Wert nur mit dem risikolosen Zins wächst (bei vollständigem Hedging darf nicht mehr erwartet werden). Mit diesem entscheidenden Ansatz folgt dann

$$V_t(t, S) + \frac{1}{2} V_{S,S}(t, S)\, S^2\, \sigma(t, S)^2 + (r(t) - D_0(t, S))\, V_S(t, S)\, S - r(t)\, V(t, S) = 0 \tag{20}$$

die berühmte Black-Scholes Partielle Differenzialgleichung (backward parabolic PDE), für die im Jahre 1997 der Nobel Preis für Wirtschaftswissenschaften (an Scholes und Merton) vergeben wurde.

Ausserdem gilt zusätzlich die Nebenbedingung (für einen Call bzw. Put)

$$\begin{aligned} &\text{Call}: \; V(T, S) = \text{Max}(S - k, 0) \\ &\text{Put}: \; V(T, S) = \text{Max}(k - S, 0) \end{aligned} \tag{21}$$

Nur wenn der Optionspreis V dieser Differenzialgleichung folgt, kann jedes Risiko aus dem Portfolio eliminiert werden (d.h., dass der Wert des Portfolio X(t) zeitlich konstant ist).

Lösen der Black-Scholes PDE (für eine Call Option)

Im nächsten Schritt gilt es nun, diese PDE zu lösen. Mit der Substitution $a = r - D$ und $b = r$ resultiert die mit BSPDE bezeichnete Black-Scholes PDE (a, b, σ seien (voläufig) unabhängig von der Zeit und dem Aktien Preis) mit der Nebenbedingung (TC, terminal condition) für eine Call Option:

$$\text{BSPDE} = \frac{1}{2} S^2 \frac{\partial^2 V(t, S)}{\partial S^2} \sigma^2 + a\, S \frac{\partial V(t, S)}{\partial S} + \frac{\partial V(t, S)}{\partial t} - b\, V(t, S) = 0;$$
$$\text{TC} = V(T, S) = \text{Max}(0, S - k);$$

Diese PDE sieht recht kompliziert aus, kann jedoch mit folgenden Transformationen und Ersetzungsregeln

$$\text{subRules} = \left\{V \to \text{Sub1},\ \frac{1}{2}(T - t)\, \sigma^2 \to \tau,\ \log\left(\frac{S}{k}\right) \to x\right\};$$
$$\text{Sub1(t\_, S\_)} := k\, v(\tau(t), x(S));$$
$$x(\text{S\_}) := \log\left(\frac{S}{k}\right);$$
$$\tau(\text{t\_}) := \frac{1}{2}(T - t)\, \sigma^2;$$

umgewandelt werden (*Mathematica* erledigt die Arbeit für uns):

```
BSPDE2 = Simplify[BSPDE //. subRules]
TC2 = FullSimplify[TC //. subRules /. S → k Exp[x]]
```

$$k\,(2\, b\, v[\tau, x] + (-2\, a + \sigma^2)\, v^{(0,1)}[\tau, x] + \sigma^2\, (-v^{(0,2)}[\tau, x] + v^{(1,0)}[\tau, x])) == 0$$

$$\text{Max}[0,\ (-1 + e^x)\, k] == k\, v[0, x]$$

Diese constant coefficient PDE kann mit dem Ansatz

$$v[\tau\_, x\_] := u[\tau, x]\, e^{x\, \alpha + \beta\, \tau};$$

in eine solche Form gebracht werden

$BSPDE3 = FullSimplify[BSPDE2]$
$TC3 = TC2$

$e^{x\alpha+\beta\tau}\,k\,((2b-2a\alpha+(\alpha-\alpha^2+\beta)\sigma^2)\,u[\tau,x]+(-2a+\sigma^2-2\alpha\sigma^2)\,u^{(0,1)}[\tau,x]+\sigma^2\,(-u^{(0,2)}[\tau,x]+u^{(1,0)}[\tau,x]))==0$

$Max[0,(-1+e^x)\,k]==e^{x\alpha}\,k\,u[0,x]$

dass mit geschickter Wahl der Parameter α und β die Faktoren von u(τ,x) und $u^{(0,1)}$, nämlich

$fac1 = Coefficient[BSPDE3[[1]], u(\tau, x)][[3]]$

$fac2 = Coefficient\left[BSPDE3[[1]], \frac{\partial u(\tau, x)}{\partial x}\right][[3]]$

$2b-2a\alpha+(\alpha-\alpha^2+\beta)\sigma^2$

$-2a+\sigma^2-2\alpha\sigma^2$

null werden. Das Nullsetzen dieser zwei Faktoren führt auf folgende Lösungen für α und β

$\alpha\beta Sub = Simplify[Solve[\{fac1 == 0, fac2 == 0\}, \{\alpha, \beta\}][[1]]]$

$\left\{\beta\to-\frac{1}{4}-\frac{a^2}{\sigma^4}+\frac{a}{\sigma^2}-\frac{2b}{\sigma^2},\ \alpha\to\frac{1}{2}-\frac{a}{\sigma^2}\right\}$

α und β oben eingesetzt ergibt für die PDE und die TC

$BSPDE4 = Simplify[BSPDE3 /. \alpha\beta Sub]$
$TC4 = (TC3 /. \alpha\beta Sub)$

$e^{x\left(\frac{1}{2}-\frac{a}{\sigma^2}\right)+\left(-\frac{1}{4}-\frac{a^2}{\sigma^4}+\frac{a}{\sigma^2}-\frac{2b}{\sigma^2}\right)\tau}\,k\,\sigma\,(u^{(0,2)}[\tau,x]-u^{(1,0)}[\tau,x])==0$

$Max[0,(-1+e^x)\,k]==e^{x\left(\frac{1}{2}-\frac{a}{\sigma^2}\right)}\,k\,u[0,x]$

was durch Multiplikation mit einem Faktor noch vereinfacht werden kann

$MaxProperty = \{a\_\ Max[0, b\_] \to Max[0, a\,b] /; a \geq 0\};$

$BSPDE5 = \left(\frac{\#1}{e^{x\left(\frac{1}{2}-\frac{a}{\sigma^2}\right)+\left(-\frac{1}{4}-\frac{a^2}{\sigma^4}+\frac{a}{\sigma^2}-\frac{2b}{\sigma^2}\right)\tau}\,k\,\sigma}\ \&\right) /@\ BSPDE4$

$TC5 = \left(\frac{\#1}{e^{x\left(\frac{1}{2}-\frac{a}{\sigma^2}\right)}\,k}\ \&\right) /@\ TC4 /. MaxProperty$

$u^{(0,2)}[\tau,x]-u^{(1,0)}[\tau,x]==0$

$\left(Max\left[0, e^{-x\left(\frac{1}{2}-\frac{a}{\sigma^2}\right)}(-1+e^x)\right] /;\ \frac{e^{-x\left(\frac{1}{2}-\frac{a}{\sigma^2}\right)}}{k}\geq 0\right)==u[0,x]$

Diese Differenzialgleichung hat nun die gewünschte Form und zwar die Form einer Wärmeleitungs- bzw. Diffusionsgleichung. Der folgende Plot zeigt auch, dass die Nebenbedingung TC exponentiell ansteigt.

```
TC5⟦1, 1⟧
Plot[TC5⟦1, 1⟧ /. {a → 0.05, σ → 0.5}, {x, -3, 3}, PlotStyle → {AbsoluteThickness[
    3], RGBColor[1, 0, 0]}, AxesLabel → {"x", "u[0,x]"}];
```

$$\mathrm{Max}\left[0,\ e^{-x\left(\frac{1}{2}-\frac{a}{\sigma^2}\right)}\left(-1+e^{x}\right)\right]$$



Die Aufgabe ist es nun, diese Wäremeleitungsgleichung zu lösen. Dies ist nicht schwierig und gehört zum Standardrepertoire jedes Physikers. Beachte, dass die Integration nur über den Bereich geführt wird, wo die Funktion ungleich Null ist.

```
ψ[x_] = TC5[[1, 1, 2]]
BSSolu[τ_, x_] = 1/(2 √(π τ)) Integrate[e^(-(ξ-x)^2/(4 τ)) ψ[ξ], {ξ, 0, ∞}, Assumptions →
    τ > 0] // ExpandAll // Simplify
```

$$e^{-x\left(\frac{1}{2}-\frac{a}{\sigma^2}\right)}\left(-1+e^{x}\right)$$

$$\frac{1}{2}\, e^{-\frac{(-2a+\sigma^2)(2x\sigma^2+2a\tau-\sigma^2\tau)}{4\sigma^4}}\left(-1+e^{x+\frac{2a\tau}{\sigma^2}}-\mathrm{Erf}\left[\frac{x\sigma^2+2a\tau-\sigma^2\tau}{2\sigma^2\sqrt{\tau}}\right]+e^{x+\frac{2a\tau}{\sigma^2}}\,\mathrm{Erf}\left[\frac{x\sigma^2+2a\tau+\sigma^2\tau}{2\sigma^2\sqrt{\tau}}\right]\right)$$

wobei Erf[x] das Integral der Gauss Verteilung ist $\mathrm{Erf}(z) = \frac{2}{\sqrt{\pi}} \int_0^z e^{-t^2}\, d\,t$.

Als Beispiel soll der Verlauf von u[x] zur Zeit $\tau == 0.5$ (Jahr) geplottet werden:

```
Plot[BSSolu[τ, x] /. {a → .05, σ → .5, τ → .5}, {x, -2, 2}, PlotStyle →
  {AbsoluteThickness[3], RGBColor[1, 0, 0]}, AxesLabel → {"x", "u[0.5,x]"}];
```



Als letzter Schritt zur Bestimmung des Option Preises muss die Lösung u[t,x] wieder zurücktransformiert werden. Gemäss den durchgeführten Transformationen

$$V = k\, v(\tau(t),\ x(S))$$

$$v(\tau,\ x) = u(\tau,\ x)\, e^{x\alpha+\beta\tau}$$

ist die Lösung VC für eine (Call) Option also

```
v[τ_, x_] = BSSolu[τ, x] e^(x α+β τ) /. αβSub

VC[t_, S_, T_, k_, a_, b_, σ_] = Simplify[k v[τ[t], x[S]]]
```

$$-\frac{1}{2} e^{b(t-T)} \left(k - e^{a(-t+T)} S + k \operatorname{Erf}\left[\frac{-(t-T)(2a-\sigma^2) + 2\operatorname{Log}[\frac{S}{k}]}{2\sqrt{2}\sqrt{(-t+T)\sigma^2}}\right] - e^{a(-t+T)} S \operatorname{Erf}\left[\frac{-(t-T)(2a+\sigma^2) + 2\operatorname{Log}[\frac{S}{k}]}{2\sqrt{2}\sqrt{(-t+T)\sigma^2}}\right]\right)$$

Der einfachste Fall, wenn keine Dividenden bezahlt werden ($D_0 = 0$), führt auf $a = b = r$. In diesem Fall erhält man die einfachere (ein Input Parameter weniger) Black-Scholes Formel

```
VC[t_, S_, T_, k_, r_, σ_] = Simplify[VC[t, S, T, k, r, r, σ]]
```

$$\frac{1}{2} \left(-e^{r(t-T)} k + S - e^{r(t-T)} k \operatorname{Erf}\left[\frac{-(t-T)(2r-\sigma^2) + 2\operatorname{Log}[\frac{S}{k}]}{2\sqrt{2}\sqrt{(-t+T)\sigma^2}}\right] + S \operatorname{Erf}\left[\frac{-(t-T)(2r+\sigma^2) + 2\operatorname{Log}[\frac{S}{k}]}{2\sqrt{2}\sqrt{(-t+T)\sigma^2}}\right]\right)$$

mit dem dazugehörigen Plot.

```
Plot[VC[0, S, .01, 120, .05, .5], {S, 80, 170}, PlotRange → All, PlotStyle →
   {AbsoluteThickness[3], RGBColor[1, 0, 0]}, AxesLabel → {"S", "V"}];
```



Als Kontrolle soll diese Lösung in die Black-Scholes Differenzialgleichung eingesetzt werden.

```
BSPDE /. V → CV // Simplify

True
```

Mathematica verifiziert, dass die gefundene Lösung in der Tat die PDE erfüllt.

Lösung der Black-Scholes PDE (für eine Put Option)

Im vorherigen Abschnitt wurde Schritt für Schritt die Lösung für eine Call Option hergeleitet. In ähnlicher Weise kann die Lösung für eine Put Option (sie hat die gleiche PDE, aber eine andere Randbedingung) gefunden werden. Aus Platzgründen wird darauf verzichtet. Es wird nur die Lösung angegeben und gezeigt, dass sie die PDE und die Randbedingung erfüllt.

$$\text{BSPDE} = \frac{1}{2} S^2 \frac{\partial^2 V(t,S)}{\partial S^2} \sigma^2 + a\, S \frac{\partial V(t,S)}{\partial S} + \frac{\partial V(t,S)}{\partial t} - b\, V(t,S) = 0;$$

$$\text{VP[t\_, S\_]} = \frac{1}{2} e^{b(t-T)} \left(k \operatorname{erfc}\left(\frac{2\log(\frac{S}{k}) - (t-T)(2a-\sigma^2)}{2\sqrt{2}\sqrt{(T-t)\sigma^2}}\right) - e^{a(T-t)} S \operatorname{erfc}\left(\frac{2\log(\frac{S}{k}) - (t-T)(\sigma^2+2a)}{2\sqrt{2}\sqrt{(T-t)\sigma^2}}\right)\right);$$

```
BSPDE /. V → VP // Simplify

True
```


Put–Call Parity

Nachdem die Formeln für die Call und Put Option gefunden wurden, ist es einfach die sogenannte Put-Call Parität

$$VC(t, S) - VP(t, S) == S - k\, e^{-r\,(T-t)} \tag{22}$$

zu verifizieren (mit Call Option Wert (VC), Put Option Wert (VP), Stock Preis (S), exercise Preis (k), Zinssatz (r) und time to expiry (T-t) . Nämlich

Allgemeine Form mit Dividende

```
equationPutCall = (VC[t, S] - VP[t, S]) // FullSimplify
```

$e^{b\,(t-T)}\,(-k + e^{a\,(-t+T)}\,S)$

Ohne Dividende

```
equationPutCall /. {a → r, b → r} // Simplify
```

$-e^{r\,(t-T)}\,k + S$

Sensitivität Analyse

Bei Vorliegen der Formel für die Call Option können auch einfach die sogenannten "Greeks"

Delta (δ)	∂_S VC	Ableitung nach dem Aktienpreis
Gamma (γ)	$\partial_{\{S,2\}}$ VC	Zweite Ableitung nach dem Aktienpreis
Rho (ρ)	∂_r VC	Ableitung nach dem Zinssatz
Theta (θ)	∂_t VC	Ableitung nach der Zeit
Vega	∂_σ VC	Ableitung nach der "Volatilität"

berechnet werden.

Stellvertretend sollen **Gamma** und **Vega** geplottet werden:

```
Plot[Evaluate[∂{S,2} VC[1/12, S, 1/6, 100, 0.05, 0.5]], {S, 0, 200}, AxesLabel →
  {S, γ}, PlotRange → All];
```



```
Plot[Evaluate[∂σ VC[1/12, 100, 1/6, 100, 0.05, σ]], {σ, 0.1, 1}, AxesLabel →
  {σ, "Vega"}, PlotRange → All];
```



Zeitabhängige Parameter

Falls die Parameter a (Zinssatz minus Dividende), b (Zinssatz) sowie σ (Volatilität) zeitabhängig sind, müssen die gemittelten Werte in die Black-Scholes Formel eingesetzt werden.

$$\texttt{Substitution} = \left\{a \to \frac{\int_t^T a[\tau]\,d\tau}{T-t},\ b \to \frac{\int_t^T b[\tau]\,d\tau}{T-t},\ \sigma^2 \to \frac{\int_t^T \sigma[\tau]^2\,d\tau}{T-t}\right\};$$

Der so resultierende Ausdruck erfüllt die Black-Scholes PDE mit variablen Koeffizienten. (wird hier nicht gezeigt).

Computer Experimente

Mit den hergeleiteten Formeln kann jetzt auch experimentiert werden,indem die verschiedenen Parameter unterschiedlich gesetzt werden.

Verhalten des Options Preises versus die Zeit.

Mit der folgenden Animation (NB Datei; die PDF Datei enthält nur das erste Bild der Animation) kann das Verhalten der Call Option versus die Zeit schön verfolgt werden.

```
With[{k = 40, T = 6/12}, (Plot[{Max[0, S - k], VC[#1, S, T, k, .07, .5]}, {S, 0,
     2 k},
     PlotRange → {0, k}, AxesLabel → {"S", "Wert"}, PlotLabel → "t = " <>
       ToString[NumberForm[#1, 3]] , PlotStyle → {Red, Blue}] &) /@ Range[
   0, T, T/20.01]];
```



Vergleich der Aktienpreis- und der Optionspreisentwicklung.

Das untenstehende Beispiel zeigt (für einen Monte-Carlo Versuch des Aktienpreises) das Verhältnis aus dem Preis und dem Anfangspreis für eine Call Option (rot), eine Aktie (grün) und eine Put Option (blau). Die Hebelwirkung und das grosse Risiko für die Optionen ist offensichtlich.


1
2
3
4
5
6
0.05 0.1 0.15 0.2 0.25 0.3